

Intusoft Newsletter

Personal Computer Circuit Design Tools

February 1994 Issue



Copyright © Intusoft, All Rights Reserved

(310) 833-0710 Fax (310) 833-9658

Interactive ICAP/4Windows & NT


I CAP/4Windows is a totally new simulation system. It integrates the interactive IsSPICE4 simulator with the SPICENET schematic entry tool, and the INTUSCOPE data processor. Together they transform how analog and mixed signal circuit simulations are performed. Toss aside your old batch style SPICE programs. The new ICAP systems are completely INTERACTIVE!

ICAP/4Windows gives you simulation features not found on more expensive simulators or on workstation products. In the November newsletter (page 33-2) we reviewed some of the new features of ICAP/4Windows including the Interactive Schematic Cross-Probing, Device Sweeping, advanced Simulation Control dialog, and powerful IsSPICE4 Simulation Scripts. Here, we will examine how simulations are run directly from the schematic and some of the new features in INTUSCOPE.

continued on page 34-2

In This Issue

- 2 **New Interactive SPICE, ICAP/4 Windows & NT**
- 5 Modeling Photodiodes
- 6 **Modeling Tubes**
- 12 The Modeling Corner:
More Asia-Pacific Parts
Using SPICE output as Stimulus



New *Interactive* ICAP/4Windows

continued
from page
34-1

As a simple example we will use the low-pass filter circuit shown in Figure 1 where IS_{PICE4} is used to perform an AC analysis to find the frequency response. Note: IS_{PICE4}, included in ICAP/4Windows, performs all of the analyses, and contains the same element support as IS_{PICE3}.


Interactive Component Variation

Through advanced inter-process communication, ICAP/4Windows allows you to change parameter values and run a simulation without leaving the schematic. Referring to Figure 1, when the new Alter tool is selected you can click on any device. The Alter dialog will display the component's current value. A new value can be entered and any simulation run by simply clicking the appropriate analysis button. Any cross-probed waveforms will be updated as the simulation runs.

In addition to the real time waveforms, you can display exact values of virtually any circuit parameter. The Measurements dialog, shown in Figure 2 displays several variables including device power, currents, and BJT base-emitter voltage.

INTUSCOPE, Data Processing

INTUSCOPE is a 32-bit program that uses extended memory. This allows INTUSCOPE to view many large waveforms without running out of memory. Several waveform scaling improvements






Figure 2, The Interactive Measurements dialog allows you to watch critical circuit parameter values such as voltage and power during the simulation.

have been made including: linking of X and Y scales for all waveforms on a graph, tiling of waveforms, and a new algorithm for better offset calculations. In Figure 3, the waveforms were linked by selecting one waveform and then selecting the Link Y Scales function. Waveforms can be unlinked or re-linked at any time.

Another major improvement is in the area of data access. INTUSCOPE can access all of the voltages, currents,

and device power dissipations directly from IsSPICE4. Using the new IsSpice WFMS... menu shown in Figure 4 (right), INTUSCOPE can display any waveform generated from the current analysis. By changing the active plot (Figure 4 left) waveforms from other analyses or past simulations can be displayed. For the AC analysis, all forms of data (magnitude, phase, db, etc.) are available.




Figure 3, INTUSCOPE now includes new “Link X/Y” scaling functions, as well as a waveform tiling feature (inset). Shown above are the low-pass filter’s frequency and phase responses for different temperatures.




Figure 4, The INTUSCOPE Plots dialog (left) allows you to select data from different analyses. The IsSpice WFMs dialog (right) provides access to circuit voltages, currents, and device parameters like power dissipation.

Third Party Interface to IsSPICE4

Intusoft has published a interface specification that will allow third party schematic entry programs to drive the simulation interactively. The specification includes a well-documented API for Windows, Windows NT (PC, DEC Alpha, and MIPS), and the Macintosh, plus several code examples including sample code for the interactive cross-probing feature.

Several major schematic vendors are adopting this standard. Once available the third party vendor's schematic will be able to interactively drive IsSPICE4 just like SPICENET. This includes the interactive cross-probing and device altering features. The specification is available to interested parties. Please inquire directly with Intusoft or one of our local distributors.

There are ICAP/4NT versions for Windows NT running on the PC, DEC Alpha, and MIPS platforms.


Modeling Photodiodes by Jeff Robson

Modeling photodiodes is fairly easy with `IsSPICE3`. The subcircuit, shown in Figure 5, is composed of a diode and a dependent source. The diode model was created by `SPICEMOD`, a program that makes SPICE models from data sheet parameters, even ones with a minimum amount of data. Since there is very little data normally available with photodiodes, this capability comes in handy. The coefficients in the G2 polynomial model the dark current and the change in spectral photosensitivity. The second subcircuit uses the behavioral features in `IsSPICE3` to add a time varying relative spectral sensitivity feature. When exposed to light, a photo diode generates a current in the pA to μ A range which is proportional to the light intensity. A simple circuit to convert this low level current to a usable voltage is shown in Figure 6.

```
.SUBCKT BPW34 4 1 3
*      Input Pwr(W) Cathode Anode
D2 6 1 DBPW34
G2 1 6 POLY(2) 1 6 4 0 0 200P 477M
* Dark current as a function of VR (200P*V(1,6))
* and Spectral Photosensitivity (A/W) at 670nm (620M*77% = 477M)
V4 6 3 ; Photodiode Current
.MODEL DBPW34 D (IS=117P RS=.132 N=2.41 BV=32 IBV=.15U
+ CJO=72P VJ=.75 M=.5 TT=504N) ; 32V 50mA diode made by SPICEMod
.ENDS
*****
.SUBCKT PHOTO 4 1 3 5 {IDARK=200P SPECP=620M}
*      Input Pwr(W) Cathode Anode Spectral Photosensitivity
D2 6 1 DMOD
BG2 1 6 I = (V(1)-V(6)) * {IDARK} + V(4) * V(5) * {SPECP} * .01
* IDARK = Dark current as a function of VR (slope)
* SPECP = Spectral Photosensitivity (A/W)
* V(5) = % reduction at target wavelength, could be made a constant if desired
V4 6 3 ; Photodiode Current
.MODEL DMOD D ; <-- Place Diode Model Here
.ENDS
```

The input to the model is a voltage analog equal to the optical power in watts

Figure 5, SPICE 2 compatible subcircuit for a photodiode. The top model is for a Siemens BPW34. The bottom circuit is a generic `IsSPICE3` version with the relative spectral density percentage as an input voltage.



Modeling Vacuum Tubes

In the February 1989 *Intusoft Newsletter*, Intusoft published one of the first vacuum diode/triode tube models using SPICE 2G syntax. The models, described in [1], used the polynomial features of IsSPICE 1.41 to simulate the nonlinear V-I response. Since then electronic designers of audio and high voltage circuits who need electronic tubes have, until now, been requesting more “serious” tubes models. In order to satisfy the development requirements and evolution of high voltage amplifiers, as well as the need for better models, EXCEM, the Intusoft distributor in France, has developed much more advanced and accurate models. They are compatible with IsSPICE3 and the new IsSPICE4 simulators, which are based on Berkeley SPICE 3F.5.

Both specific and generic models, which can be modified by the user, are available. As an example, models for the 12AU7 triode and the EL9000 pentode are given. These components are used in some simple amplifiers. The plots obtained through simulation accurately describe the behavior of the tubes. Several other tube models are included on the newsletter floppy disk for subscribers. In this issue we will deal with the triode, leaving the pentode for the next newsletter.

A Realistic Triode Model

Until now, the Intusoft triode model used a fundamental law to describe the variation of the anode current (I_a) with anode (V_a) and grid voltage (V_g):

$$I_a = K \left(V_g + \frac{V_a}{\mu} \right)^{1.5} \quad \text{Eq.1}$$

where K is the perveance and μ is the amplification factor. With this law, the anode current is continuously increasing with anode voltage. Triode tube users know that such a law is not entirely realistic. With high currents (pulsed), or when the tube is “pumped”, there is a saturation phenomenon such that the current value asymptotically reaches a maximum.

Table 1 shows the models for the 12AU7 triode, the generic triode, and the heater model. The generic triode, called TRIO1, contains equations using the IsSPICE3 B element with its If-Then-Else syntax (IF expression <> expression THEN(?) expression ELSE (:) expression). TRIO1 describes the triode's electrical phenomena while the HEAT1 subcircuit describes the thermal phenomena for the filament and cathode. TRIO1 is

Table 1, Forward and reverse conditions are treated in the triode model, as well as saturation. The sample tube, a 12AU7A, is composed of the TRIO1 and heater subcircuits. The model for the heater gives a voltage, ISAT, that is an analog of the saturation current of the cathode. All tube models COPYRIGHT EXCEM, 1993.

```
.SUBCKT T12AU7A1 1 2 3 4 5
*           Anode Grid Cathode F F'
X1 1 2 3 10 TRIO1 {SFS=0.7 VBIG=-0.9 VBIA=-1.3 MU=17 RMU=0.5 VMU=-20
+ SFMU=1.6 K=827E-6 RK=0.08 VK=-20 SFK=1.6 SIGMAG=0.05 ALPHAG=5.2 SFG=3.5}
X2 4 5 10 HEAT1 {INOM=0.15 VNOM=6.3 LAMBDA=1 RCOOL=3 TCTE=10 TNOM=1150
+ INIT=1 W=2.045 ISAT=0.099}
C2 1 2 1.5P
C3 3 1 0.5P
C4 2 3 1.6P
C5 3 4 4P
C6 3 5 4P
.ENDS
*****
.SUBCKT TRIO1 A G C ISAT
B1 15 0 V=V(G) - V(C) < -1P ?
+ {K} * (1 + {RK}) * (V(G) - V(C) / {VK})^{SFK} / (1+(V(G) - V(C) / {VK})^{SFK}) : {K}
B2 16 0 V=V(G) - V(C) < -1P ? {MU} * (1+{RMU}) * (V(G) - V(C) /
+ {VMU})^{SFMU} / (1+(V(G) - V(C) / {VMU})^{SFMU}) : {MU}
B4 9 0 V = V(G) - V(C) - {VBIG} + (V(A,C) - {VBIA}) / (V(16) + 1U)
B6 10 0 V = V(9) > 0 ? V(15) * V(9)^1.5/(V(ISAT) + 1P) : 0
B7 12 0 V = V(10) < {SFS} ? V(10) * (V(ISAT) + 1P) :
+ (V(ISAT) + 1P) * ({SFS} + (V(10) - {SFS}) * {1-SFS}) / ((1-2 * SFS) + V(10))
B8 14 0 V = V(A) - V(C) > {VBIA + 0.1M} ? (V(A) - V(C) - {VBIA}) / {ALPHAG} : 2P
B9 28 0 V = V(G) - V(C) > {VBIG + 0.1M} ? V(14) > 1P ? ((V(G) - V(C) - {VBIG}) +
+ {SIGMAG}^{1/SFG}) * V(14) / (V(G) - V(C) - {VBIG} + V(14))^{SFG} : 0
B10 8 0 V = V(G) - V(C) < 0 ? V(28) * (({VBIG+10U} + V(C) - V(G)) / {VBIG+10U}) : V(28)
B15 G C I = V(8) * V(12)
B17 A C I = (1 - V(8)) * V(12)
.ENDS
*****
.SUBCKT HEAT1 F F' ISAT
V1 F 4 0
R1 5 4 0.01
B1 5 F' V = V(7) > 0 ? I(V1) * ({VNOM / INOM - RCOOL}) *
+ (1 + {LAMBDA}) * (V(7) - 1) + {RCOOL}) : I(V1) * {VNOM / INOM}
B2 6 0 V = (V(F) - V(F')) * I(V1) > 0 ? (V(F) - V(F')) * I(V1) / {VNOM * INOM} : 1
R2 6 7 1
C1 7 0 {TCTE} IC={INIT/100}
E1 13 0 7 0 {TNOM}
B3 ISAT 0 V = V(13) > 0 ?
+ {ISAT} * V(7)^2 * EXP({W/0.8417E-4}) * (1 / {TNOM} - 1 / (V(13) + 1)) : {ISAT}
.ENDS
```

See Table 2 for notes on the B elements and parameters

connected to the anode, grid, and cathode of the triode. A 4th node (node 10) receives the saturation current created by the HEAT1 subcircuit (the triode saturation current is determined by the cathode temperature).

It was interesting to model the differences in the concavities of the IA versus VA curves for different values of the grid voltage, which are not described with Eq.1. To take this effect into account, the model introduces correction factors to the K and μ coefficients when VG is very negative. These corrections appear in the TRIO1 subcircuit where B1 and B2 describe the

Table 2, The generic parameters for the triode and heater subcircuits reveal the tremendous versatility of the model.

The Triode Parameters Are:

SFS	Shape factor of the saturation law.
VBIG	Contact potential of the grid (the voltage above which current grid may start to flow).
VBIA	Contact potential of the anode.
MU	Amplification factor at slightly negative grid voltage.
RMU	Reduction factor for MU at very negative grid voltage.
VMU	Grid voltage for mid-range MU (negative).
SFMU	Shape factor for MU reduction law.
K	Perveance at slightly negative grid voltage.
RK	Perveance reduction factor at very negative grid voltage.
VK	Grid voltage for mid-range perveance (negative).
SFK	Shape factor for perveance reduction law.
SIGMAG	Effective cross-section of the grid relative to the anode.
ALPHAG	Grid current amplification factor.
SFG	Shape factor of the grid current law.

Model Notes: B7 contains an arbitrary saturation law modeled by the shape factor SFS to match the available data. SFS should be between 0 and 1, and the lower SFS, the sloppier the saturation law. V(15) is the effective perveance. V(16) is the effective MU. TRIO1 describes only the static behavior of the triode and neglects secondary emission (which would occur at high VG and low VA).


The Heater Parameters Are:

INOM	The nominal heater current at nominal voltage.
VNOM	The nominal heater voltage (causing nominal temperature).
LAMBDA	Temperature coefficient of the heater resistance. (normalized to the nominal temperature)
RCOOL	Resistance of the cold heater.
TCTE	The time constant for the heater temperature.
TNOM	The nominal heater temperature in K.
INIT	Initial heater temperature in % of TNOM
W	Work function of the heater, in eV.
ISAT	The saturation current at nominal heater voltage.

Model Notes: B1 delivers the power received by the heater. Only conductive dissipation is considered (no radiative dissipation and back scattered electrons at RF don't heat the cathode), therefore V(7) is a normalized temperature. B4 contains the Richardson-Duschman law, with an exponential containing $B=W/k$, the Boltzmann's constant being $0.8617e-4$ eV/K for Tungsten (see [4]) $W=4.5$ eV. It may be 2.2 times lower for an oxide-coated cathode (see [3] p 173). B3 delivers the saturation current.


variations of the perveance and amplification factor. The values of B1 and B2 are then used in B4 and B6 which describe the law [3]. B7 takes into account the saturation variations which are seen when the cathode current is very high. B8, B9, and B10 describe the behavior of the grid current when the grid is positive with respect to the cathode. Reasonable grid current behavior is very important for high power applications. B15 and B16 use the values of the aforementioned B elements to establish the currents in the tube terminals.

Figure 7 shows the curve family obtained for the 12AU7 triode. Note, the proper behavior of the current for negative and positive grid voltages. The saturation current value (about 100mA) for nominal voltages and heating currents was fixed



arbitrarily; this value is actually dependent on the manufacturing process and tube ageing. The saturation current, I_{sat} , depends on the voltage and heating current. The corresponding heater model is based on thermal considerations and on the Richardson-Duschman law. It is, however, possible (and easy) to suppress this effect, which consumes a large amount of simulation time for little return; in this case the filament (HEAT1) is replaced by a resistor and a voltage source giving I_{sat} , in the subcircuit T12AU7A.

We can see in Figure 8 that the IsSPICE4 cross-probed waveforms match pretty well with the real behavior of the triode. An input sinusoid (IK) of (12V) was used resulting in a peak anode current of 52mA.






Figure 9, An amplifier combining a power MOSFET with the new triode tube model for a 12AU7A. Waveforms are available either on the schematic or in INTUSCOPE.

A Triode Amplifier

Figure 9 shows an amplifier schematic using a MOS transistor and a double triode (used in a cascode stage), in order to obtain, a high impedance, a very high voltage gain, a high level of feedback, and a good frequency characteristic. The cross probed waveforms show the output voltage, for an input voltage at V(13). In order to speed up the simulation the heater model was replaced with a resistor and voltage source as noted on the previous page. The MOS model is itself a sophisticated subcircuit found it in the Intusoft library. At the 100kHz frequency of the 1V input signal, the output is about 278 V. The curve of the Figure 10 gives the frequency response of the amplifier for small signals.




Figure 10, An INTUSCOPE graph of V(4), the small signal response of the tube amplifier shown in Figure 7.

Part II, Conclusions, and References

In the next issue of the *Intusoft Newsletter* we will continue the work started here. A generic and specific model for a pentode will be given along with its characteristic curves. A high voltage amplifier using the pentode will also be simulated. This work vividly shows the extent to which the behavioral capabilities of IsSPICE3 and IsSPICE4 can reach. Clearly, complex mathematical models of other electronic devices are possible.

References

[1] L.G. Meares, C.E. Hymowitz, "SPICE APPLICATIONS HANDBOOK", Intusoft, 1990, NL-12 February 1989

[2] Frederic Broyde, "Modelisation Et Simulation Des Circuits A Tubes Avec IsSpice3", Electronique Radio-Plans, Oct. 1993

[3] Frederick E. Terman, "Electronic and Radio Engineering", McGraw-Hill, 1955

[4] N. Ashcroft, N. Mermin, "Solid State Physics", Cornell University, 1976

The Tube Models were created by:
EXCEM 12, Chemin des Hauts de Clairefontaine
78580 MAULE FRANCE
Tel 33 (1) 34 75 13 65, Fax 33 (1) 34 75 13 66

EXCEM provides worldwide services in the fields of electronics R&D, with a focus on RF, theoretical and practical EMC, electronic simulations and DSP. They are also the Intusoft representative in France. Among other things Excem manufactures high-voltage wide-band tube amplifiers for driving high impedance loads (e.g. short E-field antennas).

Intusoft NEWS: CompuServe Information Service

Intusoft maintains a bulletin board service under the CADD/CAM/CAE Vendor forum on CompuServe. A variety of technical articles, application notes, demos, and free SPICE models are posted there. To connect with Intusoft you can navigate through CompuServe's computing support menus to reach the CADD/CAM/CAE Vendor forum or type "Go CADDVEN" at any ! point prompt. Then select the "All CADD/CAM/CAE" section. The CompuServe address is **intusoft@compuserve.com**. Internet users can also send e-mail messages to Intusoft using the address format: "info@intusoft.com".

Interesting items posted by Intusoft include:

- **Multidec**, a utility to create SPICE models of multiple coupled lossy transmission lines
- SPICE models, from Philips Semiconductors, for ALS, FAST, ABT, and 3V BiCMOS logic families (with slow, typical, and fast variations)
- Berkeley SPICE compatible models for IGBTs, fuses, and dual-gate MOSFETs
- Application Note called "Solving SPICE Convergence Problems"
- Technical Papers on modeling: IGBTs, Fuses, and RF Amplifiers/Oscillators
- Software Demos of SPICEMOD, the SPICE modeling program, and FILTERMASTER, a program for synthesizing passive LC filters

Modeling Corner

In this issue of The Intusoft Modeling Corner we bring you new models from Asia-Pacific vendors.

More Pacific Rim Parts Join Library

Intusoft tech support has created new models for BJTs and IGBTs. Enclosed on the floppy disk are over **200 models** from the following vendors: Matsushita, Fuji, and Sanyo. Thanks again to AMSC, Inc. an Intusoft representative in Japan.

```
.MODEL Q2SC1567 NPN (IS=50.7F NF=1 BF=422 VAF=180 IKF=.3 ISE=12P NE=2
+ BR=4 NR=1 VAR=16 IKR=.45 RE=.103 RB=.412 RC=41.2M XTB=1.5 CJE=110P
+ VJE=1.1 MJE=.5 CJC=35.5P VJC=.3 MJC=.3 TF=1.32N TR=921N)
* Matsushita 100V .5A 120MHz NPN BJT
.MODEL Q2SC2592 NPN (IS=101F NF=1 BF=421 VAF=241 IKF=.6 ISE=24.1P NE=2
+ BR=4 NR=1 VAR=16 IKR=.9 RE=.663 RB=2.65 RC=.265 XTB=1.5 CJE=183P
+ VJE=1.1 MJE=.5 CJC=59.1P VJC=.3 MJC=.3 TF=795P TR=552N)
* Matsushita 180V 1A 200MHz NPN BJT
.SUBCKT Q2SC2740 1 2 3
* Terminals: C B E
* Matsushita 400V 10A NPN Power BJT
Q1 1 2 3 QPWR .67
Q2 1 4 3 QPWR .33
RBS 2 4 5
.MODEL QPWR NPN (IS=368F NF=1 BF=64.8 VAF=360 IKF=3.33 ISE=703P NE=2
+ BR=4 NR=1 VAR=20 IKR=5 RE=31.3M RB=.125 RC=12.5M XTB=1.5
+ CJE=322P VJE=.6 MJE=.3 CJC=209P VJC=.22 MJC=.2 TF=14.4N TR=3.4U)
.ENDS
* Fuji 600 Volt 30 Amp 75NS N-Channel IGBT
.SUBCKT I6MBI30F 71 72 74
* TERMINALS: C G E
.MODEL SW NMOS (LEVEL=3 VTO=0 KP=5)
.MODEL QOUT PNP (IS=3.24P NF=1.2 BF=5.1 CJE=3.21N TF=75N XTB=1.3)
.MODEL MFIN NMOS (LEVEL=3 VMAX=400K THETA=53.3M ETA=2M
+ VTO=4.5 KP=1.61)
.MODEL DR D (IS=324F CJO=85.8P VJ=1 M=.82)
.MODEL DO D (IS=324F BV=600 CJO=2N VJ=1 M=.7)
.MODEL DE D (IS=324F BV=14.3 N=2)
.MODEL DLIM D (IS=100N)
.MODEL DMBI30 D (IS=3.45U RS=59.2M N=1.75 BV=600 IBV=10U
+ CJO=1.23N VJ=.75 M=.333 TT=432N)
Q1 83 81 85 QOUT
M1 81 82 83 83 MFIN L=1U W=1U
DSD 83 81 DO
DBE 85 81 DE
RC 85 71 26.2M
RE 83 73 2.62M
RG 72 82 40.1
CGE 82 83 2.83N
CGC 82 71 1P
EGD 91 0 82 81 1
VFB 93 0 0
FFB 82 81 VFB 1
CGD 92 93 1.2N
R1 92 0 1
D1 91 92 DLIM
DHV 94 93 DR
R2 91 94 1
D2 94 0 DLIM
LE 73 74 7.5N
DLV 94 95 DR 13
RLV 95 0 1
ESD 96 93 POLY(1) 83 81 19 1
MLV 95 96 93 93 SW
DFW 74 71 DMBI30
.ENDS
```

Sample Pacific Rim Models

Stimulating Circuits: Using SPICE Output

Most SPICE programs are accompanied by some sort of graphical data processing program. But few are as powerful as the INTUSCOPE program integrated into the ICAP simulation system. Post processing programs generally display SPICE output data and provide some detailed analysis capabilities such as waveform cursor measurements and the ability to multiply waveforms.

One of the unique features of INTUSCOPE is the ability to take any waveform that is displayed and digitize it for use as stimulus in another ISPACE simulation. This feature allows you to easily break the circuit into manageable sections and “feed” output waveforms from one section in to another.

As a simple example, we converted the output of an IBIS digital gate that is driving a PCI bus into an input signal. It would normally be very difficult to utilize this waveform as stimulus without this feature. The waveform shown in Figure 11 was displayed in an INTUSCOPE graph then digitized using the PWL function in INTUSCOPE. This saves the waveform in the standard SPICE PWL format that can be used with any compatible program. A resolution option is available to reduce the number of data points saved, if desired, since a large number of PWL points can slow the simulation unnecessarily.

